Solution to steady-state model

In[2]:=
```
OriginalSystem = {
   0 == 1 - ϕ^3 - γ η^3,
   0 == 2 R^2 η^3 - 3 R^2 η^2 + R^2 - Rcrit^2,
   0 == R^2 ϕ^3 + (Q^2 Rcrit^2 - R^2 (1 + 2 η^3)) ϕ + 2 η^3 R^2
  }; OriginalSystem // TableForm
```

Out[2]//TableForm=

$$0 == 1 - \gamma\,\eta^3 - \phi^3$$
$$0 == R^2 - \mathrm{Rcrit}^2 - 3\,R^2\,\eta^2 + 2\,R^2\,\eta^3$$
$$0 == 2\,R^2\,\eta^3 + \left(Q^2\,\mathrm{Rcrit}^2 - R^2\,\left(1 + 2\,\eta^3\right)\right)\,\phi + R^2\,\phi^3$$

In[3]:=
```
(* Substitute η → ϕ ρ *)
System = Simplify[OriginalSystem /. η → ϕ * ρ]; System // TableForm
```

Out[3]//TableForm=

$$\left(1 + \gamma\,\rho^3\right)\,\phi^3 == 1$$
$$\mathrm{Rcrit}^2 == R^2\,(-1 + \rho\,\phi)^2\,\left(1 + 2\,\rho\,\phi\right)$$
$$\phi\,\left(-Q^2\,\mathrm{Rcrit}^2 + R^2\,(-1 + \phi)\,\left(-1 - \phi + 2\,\rho^3\,\phi^2\right)\right) == 0$$

In[4]:=
```
(* Solve Eq. 1 for ϕ *)
AllSol1 = Solve[System[[1]], ϕ]
```

Out[4]=
$$\left\{\left\{\phi \to \frac{1}{\left(1 + \gamma\,\rho^3\right)^{1/3}}\right\},\ \left\{\phi \to -\frac{(-1)^{1/3}}{\left(1 + \gamma\,\rho^3\right)^{1/3}}\right\},\ \left\{\phi \to \frac{(-1)^{2/3}}{\left(1 + \gamma\,\rho^3\right)^{1/3}}\right\}\right\}$$

In[5]:=
```
Sol1 = AllSol1[[1]]
```

Out[5]=
$$\left\{\phi \to \frac{1}{\left(1 + \gamma\,\rho^3\right)^{1/3}}\right\}$$

In[6]:=
```
(* Solve Eq. 2 for R *)
AllSol2 = Solve[System[[2]], R]
```

Out[6]=
$$\left\{\left\{R \to -\frac{\mathrm{Rcrit}}{\sqrt{1 - 3\,\rho^2\,\phi^2 + 2\,\rho^3\,\phi^3}}\right\},\ \left\{R \to \frac{\mathrm{Rcrit}}{\sqrt{1 - 3\,\rho^2\,\phi^2 + 2\,\rho^3\,\phi^3}}\right\}\right\}$$

In[7]:=
```
Sol2 = AllSol2[[2]]
```

Out[7]=
$$\left\{R \to \frac{\mathrm{Rcrit}}{\sqrt{1 - 3\,\rho^2\,\phi^2 + 2\,\rho^3\,\phi^3}}\right\}$$

```
(* Simplify Eq. 3 into a univariate function in ρ *)
```

In[8]:=
```
Eq3 = System[[3]] /. Sol2 /. Sol1
```

Out[8]=
$$\frac{-Q^2\,\mathrm{Rcrit}^2 + \dfrac{\mathrm{Rcrit}^2\,\left(-1 + \frac{2\,\rho^3}{\left(1+\gamma\,\rho^3\right)^{2/3}} - \frac{1}{\left(1+\gamma\,\rho^3\right)^{1/3}}\right)\,\left(-1 + \frac{1}{\left(1+\gamma\,\rho^3\right)^{1/3}}\right)}{1 + \frac{2\,\rho^3}{1+\gamma\,\rho^3} - \frac{3\,\rho^2}{\left(1+\gamma\,\rho^3\right)^{2/3}}}}{\left(1 + \gamma\,\rho^3\right)^{1/3}} == 0$$

In[12]:= `Eq3v2 = Eq3[[1]] * (1 + γ ρ^3)^(1/3) == Eq3[[2]] * (1 + γ ρ^3)^(1/3)`

Out[12]=
$$-Q^2\,\mathrm{Rcrit}^2+\frac{\mathrm{Rcrit}^2\left(-1+\frac{2\rho^3}{(1+\gamma\rho^3)^{2/3}}-\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)\left(-1+\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)}{1+\frac{2\rho^3}{1+\gamma\rho^3}-\frac{3\rho^2}{(1+\gamma\rho^3)^{2/3}}}==0$$

In[16]:= `Eq3v3 = AddSides[Eq3v2, Q^2 Rcrit^2]`

Out[16]=
$$\frac{\mathrm{Rcrit}^2\left(-1+\frac{2\rho^3}{(1+\gamma\rho^3)^{2/3}}-\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)\left(-1+\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)}{1+\frac{2\rho^3}{1+\gamma\rho^3}-\frac{3\rho^2}{(1+\gamma\rho^3)^{2/3}}}==Q^2\,\mathrm{Rcrit}^2$$

In[17]:= `Eq3v4 = Assuming[Rcrit > 0, DivideSides[Eq3v3, Rcrit^2]]`

Out[17]=
$$\frac{\left(-1+\frac{2\rho^3}{(1+\gamma\rho^3)^{2/3}}-\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)\left(-1+\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)}{1+\frac{2\rho^3}{1+\gamma\rho^3}-\frac{3\rho^2}{(1+\gamma\rho^3)^{2/3}}}==Q^2$$

In[18]:= `Eq3v5 =`
`MultiplySides[Eq3v4, Denominator[Eq3v4[[1]]], GenerateConditions → Automatic]`

Out[18]=
$$\left(-1+\frac{2\rho^3}{(1+\gamma\rho^3)^{2/3}}-\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)\left(-1+\frac{1}{(1+\gamma\rho^3)^{1/3}}\right)==Q^2\left(1+\frac{2\rho^3}{1+\gamma\rho^3}-\frac{3\rho^2}{(1+\gamma\rho^3)^{2/3}}\right)$$

In[19]:= `Eq3v6 = Simplify[Eq3v5]`

Out[19]=
$$\frac{-1+(1+\gamma\rho^3)^{1/3}+\rho^3\left(-2-\gamma+2\,(1+\gamma\rho^3)^{1/3}\right)+Q^2\left(1+(2+\gamma)\,\rho^3-3\rho^2\,(1+\gamma\rho^3)^{1/3}\right)}{1+\gamma\rho^3}==0$$

In[20]:= `Eq3v7 = Expand[MultiplySides[Eq3v6, 1 + γ ρ^3, GenerateConditions → Automatic]]`

Out[20]=
$$-1+Q^2-2\rho^3+2Q^2\rho^3-\gamma\rho^3+Q^2\gamma\rho^3+(1+\gamma\rho^3)^{1/3}-3Q^2\rho^2\,(1+\gamma\rho^3)^{1/3}+2\rho^3\,(1+\gamma\rho^3)^{1/3}==0$$

In[26]:= `Eq3v8 = SubtractSides[Eq3v7, Q^2 - 2 ρ^3 + 2 Q^2 ρ^3 + Q^2 γ ρ^3 - 1 - γ ρ^3]`

Out[26]=
$$(1+\gamma\rho^3)^{1/3}-3Q^2\rho^2\,(1+\gamma\rho^3)^{1/3}+2\rho^3\,(1+\gamma\rho^3)^{1/3}==1-Q^2+2\rho^3-2Q^2\rho^3+\gamma\rho^3-Q^2\gamma\rho^3$$

In[27]:= `Eq3v9 = Expand[Eq3v8[[1]]^3 - Eq3v8[[2]]^3] == 0`

Out[27]=
$$\begin{aligned}&3Q^2-3Q^4+Q^6-9Q^2\rho^2+18Q^2\rho^3-18Q^4\rho^3+6Q^6\rho^3-2\gamma\rho^3+9Q^2\gamma\rho^3-9Q^4\gamma\rho^3+\\&\quad 3Q^6\gamma\rho^3+27Q^4\rho^4-36Q^2\rho^5-9Q^2\gamma\rho^5+36Q^2\rho^6-36Q^4\rho^6-15Q^6\rho^6-6\gamma\rho^6+\\&\quad 36Q^2\gamma\rho^6-36Q^4\gamma\rho^6+12Q^6\gamma\rho^6-3\gamma^2\rho^6+9Q^2\gamma^2\rho^6-9Q^4\gamma^2\rho^6+3Q^6\gamma^2\rho^6+\\&\quad 54Q^4\rho^7+27Q^4\gamma\rho^7-36Q^2\rho^8-36Q^2\gamma\rho^8+24Q^2\rho^9-24Q^4\rho^9+8Q^6\rho^9+\\&\quad 36Q^2\gamma\rho^9-36Q^4\gamma\rho^9-15Q^6\gamma\rho^9-6\gamma^2\rho^9+18Q^2\gamma^2\rho^9-18Q^4\gamma^2\rho^9+6Q^6\gamma^2\rho^9-\\&\quad \gamma^3\rho^9+3Q^2\gamma^3\rho^9-3Q^4\gamma^3\rho^9+Q^6\gamma^3\rho^9+54Q^4\gamma\rho^{10}-36Q^2\gamma\rho^{11}+8\gamma\rho^{12}==0\end{aligned}$$

In[28]:= `Eq3v10 = Collect[Eq3v9, ρ]`

Out[28]=
$$\begin{aligned}&3Q^2-3Q^4+Q^6-9Q^2\rho^2+\\&\quad\left(18Q^2-18Q^4+6Q^6-2\gamma+9Q^2\gamma-9Q^4\gamma+3Q^6\gamma\right)\rho^3+27Q^4\rho^4+\left(-36Q^2-9Q^2\gamma\right)\rho^5+\\&\quad\left(36Q^2-36Q^4-15Q^6-6\gamma+36Q^2\gamma-36Q^4\gamma+12Q^6\gamma-3\gamma^2+9Q^2\gamma^2-9Q^4\gamma^2+3Q^6\gamma^2\right)\rho^6+\\&\quad\left(54Q^4+27Q^4\gamma\right)\rho^7+\left(-36Q^2-36Q^2\gamma\right)\rho^8+\\&\quad\left(24Q^2-24Q^4+8Q^6+36Q^2\gamma-36Q^4\gamma-15Q^6\gamma-6\gamma^2+18Q^2\gamma^2-18Q^4\gamma^2+\right.\\&\qquad\left.6Q^6\gamma^2-\gamma^3+3Q^2\gamma^3-3Q^4\gamma^3+Q^6\gamma^3\right)\rho^9+54Q^4\gamma\rho^{10}-36Q^2\gamma\rho^{11}+8\gamma\rho^{12}==0\end{aligned}$$

In[35]:=
```
Coefs = Join[{Eq3v10⟦1⟧ /. ρ → 0}, Table[Coefficient[Eq3v10⟦1⟧, ρ^n], {n, 1, 12}]];
TableForm[Coefs]
```

Out[36]//TableForm=

$3 Q^2 - 3 Q^4 + Q^6$
0
$-9 Q^2$
$18 Q^2 - 18 Q^4 + 6 Q^6 - 2 \gamma + 9 Q^2 \gamma - 9 Q^4 \gamma + 3 Q^6 \gamma$
$27 Q^4$
$-36 Q^2 - 9 Q^2 \gamma$
$36 Q^2 - 36 Q^4 - 15 Q^6 - 6 \gamma + 36 Q^2 \gamma - 36 Q^4 \gamma + 12 Q^6 \gamma - 3 \gamma^2 + 9 Q^2 \gamma^2 - 9 Q^4 \gamma^2 + 3 Q^6 \gamma^2$
$54 Q^4 + 27 Q^4 \gamma$
$-36 Q^2 - 36 Q^2 \gamma$
$24 Q^2 - 24 Q^4 + 8 Q^6 + 36 Q^2 \gamma - 36 Q^4 \gamma - 15 Q^6 \gamma - 6 \gamma^2 + 18 Q^2 \gamma^2 - 18 Q^4 \gamma^2 + 6 Q^6 \gamma^2 - \gamma^3 + 3 Q^2 \gamma^3 - 3 Q$
$54 Q^4 \gamma$
$-36 Q^2 \gamma$
8γ

Solutions to *R* and *ϕ* in terms of *ρ*

In[40]:=
```
fϕ = ϕ /. Sol1
```

Out[40]= $\frac{1}{\left(1 + \gamma \rho^3\right)^{1/3}}$

In[42]:=
```
fR = R /. Sol2
```

Out[42]= $\frac{\text{Rcrit}}{\sqrt{1 - 3 \rho^2 \phi^2 + 2 \rho^3 \phi^3}}$

In[43]:=
```
fη = ϕ * ρ
```

Out[43]= $\rho \phi$